Maxima: istruzioni per l'uso

ver. 09.09
Enrico Centenaro

Email: enrico.centenaro@istruzione.it

Web: http://www.centenaro.net

Riassunto

Maxima[4] è un Computer Algebra System, un programma che permette di fare calcolo simbolico e numerico, grafici di funzioni in 2 e 3 dimensioni e molto altro ancora.

Queste pagine mostrano una fugace panoramica delle funzionalità di Maxima, avendo in mente un utilizzo fatto in laboratorio, in una scuola secondaria come attività complementare.

Questo documento[1] è *liberamente tratto* da un tutorial del professor Scott Hudson[2] che mi ha autorizzato a usare il suo elaborato "come meglio credo". La versione più aggiornata è reperibile all'indirizzo dell'autore[1].

Questo documento è stato redatto utilizzando il programma TeXmacs, una estensione EMACS che utilizza il magnifico ambiente LaTeX, (si vedano [3], [2], [5] e [?]). Segnalazioni di errori, suggerimenti e saluti sono molto apprezzati.

2. Professor and Electrical Engineering Coordinator School of Electrical Engineering and Computer Science, Washington State University, Tri-Cities. http://www.tricity.wsu.edu/~hudson/

Indice

1 Insiemi

Con Maxima gli insiemi si possono definire per elencazione attraverso l'istruzione set. Basta includere gli elementi che vi sono contenuti. Le operazioni che sono previste sono unione, intersezione, differenza e molte altre (vedi il manuale completo). Qui di seguito sono riportati alcuni calcoli che dovrebbero chiarire le sintassi dei vari comandi .

```
(%i12) set(); # insieme vuoto
```

(%o12) $\{\}$

```
(%i13) A:set(1,2,3); # insieme composto dai numero 1,2,3
```

(%o27) $\{1, 2, 3\}$

```
(%i28) B:set(1,3,6,7,8,9,10); # ovvio
```

(%o21) $\{1, 3, 6, 7, 8, 9, 10\}$

```
(%i22) C:intersection(A,B); # C è l'intersezione fra A e B
```

(%o22) $\{1, 3\}$

```
(%i23) D:union(A,B); # D è l'unione
```

(%o23) $\{1, 2, 3, 6, 7, 8, 9, 10\}$

```
(%i24) powerset(C); # Insieme potenza ovvero insieme delle parti
```

(%o24) $\{\{\}, \{1\}, \{1, 3\}, \{3\}\}$

```
(%i25) cartesian_product(A,C); # prodotto cartesiano
```

(%o25) $\{[1, 1], [1, 3], [2, 1], [2, 3], [3, 1], [3, 3]\}$

```
(%i26) setdifference(B,C); # differenza fra B e C
```

(%o26) $\{6, 7, 8, 9, 10\}$

```
(%i27) cardinality(B); # numero di elementi di B
```

(%o28) 7

Esempio 1. Dati gli insiemi $A = \{1, ..., 8\}$, $B = \{2, 4, ..., 20\}$ e $C = \{3, 6, ..., 18\}$, calcolare $(A \cap C) \cup (B \cap C)$.

```
(%i32) A:set(1,2,3,4,5,6,7,8);
```

(%o33) $\{1, 2, 3, 4, 5, 6, 7, 8\}$

```
(%i34) B:set(2,4,6,8,10,12,14,16,18,20);
```

(%o34) $\{2, 4, 6, 8, 10, 12, 14, 16, 18, 20\}$

```
(%i35) C:set(3,6,9,12,15,18);
```

(%o35) $\{3, 6, 9, 12, 15, 18\}$

```
(%i36) union(intersection(A,C),intersection(B,C));
```

(%o36) $\{3, 6, 12, 18\}$

Esempio 2. Verificare la proprietà commutativa della intersezione per gli insiemi A e B. Inoltre utilizzando anche C verificare la proprietà associativa.

```
(%i37) intersect(A,B);
```

(%o37) $\{2, 4, 6, 8\}$

```
(%i38) intersect(B,A);
```

(%o38) $\{2, 4, 6, 8\}$

```
(%i39) union(union(A,B),C);
```

(%o39) $\{1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 12, 14, 15, 16, 18, 20\}$

```
(%i40) union(A,union(B,C));
```

(%o40) $\{1,2,3,4,5,6,7,8,9,10,12,14,15,16,18,20\}$

Esempio 3. Supponendo A l'insieme universo, si calcoli $\bar{B}_U$ (complementare di B rispetto a U)

```
(%i41) setdifference(A,B);
```

(%o42) $\{1,3,5,7\}$

Il comando subset permette di determinare il sottoinsieme di un insieme composto dagli elementi che rendono vera una certa funzione o predicato (si veda più avanti).

Esempio 4. Si consideri l'insieme $A=\{0,2,4,6,8,9,11\}$ e si ricavi il sottoinsieme B composto dai multipli di 3.

Evidentemente un numero è divisibile per 3 se il resto della divisione per 3 è zero, cioè se il predicato $p(x)=x \operatorname{div} 3$ è vero. Con Maxima il resto si calcola con l'istruzione remainder.

```
(%i1) p(x):=is(remainder(x,3)=0)$ # definisco il predicato con il comando 'is'
(%i4) p(3); # ovvio
```

(%o4) **true**

```
(%i5) p(7); # ovvio
```

(%o5) **false**

```
(%i6) A:set(0,2,4,6,8,9,11)$
(%i7) subset(A,p); # i multipli di 3
```

(%o8) $\{0,6,9\}$

```
(%i9) partition_set(A,p); # la partizione di A usando il predicato 'p'
```

(%o9) $[\{2,4,8,11\},\{0,6,9\}]$

Adesso tocca a te. Aiutandoti con quanto abbiamo fatto fino ad ora fai pratica con Maxima eseguendo gli esercizi seguenti.

Esercizio 1. Sia U l'insiema ambiente composto dai numeri compresi fra 1 e 20 (estremi inclusi). Considerando A l'insieme dei numeri naturali dispari compresi fra 1 e 9 (estremi inclusi), B l'insieme dei numeri naturali compresi fra 5 e 15 (estremi inclusi) e $C=\{2,4,5,11,17\}$. Calcola:

a) $(A\cup C)\cap B$

b) $\bar{A}_U$

c) $\bar{B}_U$

d) $A\cap\bar{B}_U$

e) $\overline{A\cap C}_U$

f) Insieme delle parti di A

g) L'insieme di tutti i sottoinsiemi di A aventi 4 elementi.

Esercizio 2. Stabilisci se le seguenti uguaglianze fra gli insiemi dell'esercizio precedente sono vere o false.

a) $A\cup B=B\cup A$

b) $(A\cup B)\cap C=A\cup(B\cap C)$

c) $(A\cup B)\cap C=(A\cap C)\cup(B\cap C)$

2 Logica

Innanzitutto impariamo il nome degli operatori logici che Maxima offre.

Nella tabella che segue, la prima colonna mostra come gli operatori vengono rappresentati in Maxima

`not p`	negazione di p
`p and q`	congiunzione di p e q
`p or q`	disgiunzione di p e q

Come vedete non sono molti, ci auguriamo che il team di sviluppo di Maxima dedichi più attenzione a questo settore.

Facciamo qualche calcolo logico.

```
(%i9) q:true$
(%i10) p:false$
(%i11) p and q;
(%o11) false
(%i12) p or q;
(%o12) true
(%i13) not(p and q) = p or q; # legge di De Morgan
(%o13) true
```

3 Calcolo Aritmetico e Algebrico

Maxima può essere utilizzato come un potente calcolatore.

```
(%i5) 144*17 - 9;
(%o5) 2439
```

Si possono fare dei calcoli con numeri molto grandi, nel calcolo seguente valutiamo la 25-esima potenza di 144:

```
(%i6) 144^25;
(%o6) 9100438150002149773327585275342566324927152603256586 24
```

Questi calcoli non li fanno le normali calcolatrici. Adesso calcoliamo la radice 25-esima:

```
(%i7) (%o6)^(1/25);
(%o8) 144
```

L'inserimento di espressioni numeriche intere o razionali restituisce il valore semplificato il più possibile come nell'esempio che segue. Per avere i risultati in formato decimale si può usare la funzione float. In alternativa, se qualche elemento della riga di input è in formato decimale, allora il risultato sarà visualizzato ancora in decimale. Si noti come vengono utilizzati i nomi delle righe in luogo dei valori stessi.

```
(%i2 102/50+17/11;
```

(%o2) $\frac{986}{275}$

```
(%i3) sqrt(%o2);
```

(%o3) $\frac{\sqrt{986}}{5\sqrt{11}}$

```
(%i4) float(%o3);
(%o4) 1.893529652647285
```

```
(%i5) sqrt(102/50+17.0/11);
```

(%o5) 1.893529652647285

Oltre alle classiche operazioni Maxima consente di calcolare, il mcm, il MCD, il quoziente e il resto di divisioni intere; permette di verificare se un numero è primo, ne calcola la sua scomposizione e trova i suoi divisori.

Nei calcoli che seguono imparerete a utilizzare queste istruzioni.

```
(%i43) quotient(44,6);
```

(%o44) 7

```
(%i45) remainder(44,6);
```

(%o45) 2

```
(%i46) quotient(44,6)*6 + remainder(44,6);
```

(%o46) 44

```
(%i47) factor(44);
```

(%o47) $2^2\,11$

Calcoliamo il mcm e MCD, ricordo che massimo comun divisore in inglese si dice greatest common divisor (gcd), mentre il mcm in inglese si chiama least common multiple (lcm). Attenzione per utilizzare la funzione lcm bisogna "caricarla" digitando il comando load'("functs");.

```
(%i1) gcd(44,121);
```

(%o1) 11

```
(%i2) 44*121/gcd(44,121);
```

(%o9) 484

```
(%i10) primep(47);
```

(%o6) **true**

```
(%i7) primep(1234567);
```

(%o7) **false**

```
(%i8) lcm(12,3);
```

(%o2) 484)

Esercizio 3. Calcolare il mcm e MCD fra i seguenti gruppi di numeri:

a) 3300, 2625

b) 25875, 16335

c) 10500, 174150

Esercizio 4. Stabilisci se i seguenti numeri sono primi:
1777, 1234567, 9721

Vediamo qualche operazione con i simboli:

```
(%i9) (x + 2*y)^4;
```

(%o9) $(2\,y+x)^4$

```
(%i10) expand(%);
```

(%o10) $16\,y^4+32\,x\,y^3+24\,x^2\,y^2+8\,x^3\,y+x^4$

```
(%i11) factor(%);
```

(%o11) $(2\,y+x)^4$

Si noti l'uso del simbolo % per indicare la riga precedente. Come al solito attraverso un esempio introduciamo delle funzioni e mostriamo il loro utilizzo.

Esempio 5. Calcolare il valore numerico delle espressioni con i valori indicati:

$$\frac{(a+b)^2}{a+b}+\left(\frac{1}{a+b}-\frac{1}{a-b}+\frac{2a}{a^2-b^2}\right)\cdot\frac{a^2-b^2}{2}$$

per $a=\frac{1}{2}$ e $b=1$.

$$\frac{\frac{x\,(y+x)}{y^3+x^3}+\frac{x-y}{x^2-y^2}}{\left(1-\frac{x}{y+x}\right)^2}-\frac{1}{x-y}$$

per $y=-\frac{1}{2}$ e $x=1$

```
(%i3) esp:(a+b)^2/(a+b)+(1/(a+b)-1/(a-b)+2*a/(a^2-b^2))*(a^2-b^2)/2;
```

(%o3) $\dfrac{(a^2-b^2)\left(\frac{2\,a}{a^2-b^2}+\frac{1}{b+a}-\frac{1}{a-b}\right)}{2}+b+a$

```
(%i4) subst([a=1/2,b=1],%o3);
```

(%o8) 1

```
(%i9) esp2:((x*(x+y)/(x^3+y^3)+(x-y)/(x^2-y^2))/((1-x/(x+y))^2))-1/(x-y);
```

(%o9) $\dfrac{\frac{x\,(y+x)}{y^3+x^3}+\frac{x-y}{x^2-y^2}}{\left(1-\frac{x}{y+x}\right)^2}-\dfrac{1}{x-y}$

```
(%i10) subst([x=1,y=-1/2],%o9);
```

(%o11) $\dfrac{40}{21}$

Passiamo ora al calcolo letterale: operazioni con i monomi.

Esempio 6. Ridurre alla forma normale il seguente monomio:

$$\left(-\frac{1}{6}\right)x^3\left(+\frac{3}{4}\right)x^2y^3\left(-\frac{2}{5}\right)\mathrm{xy}^4z$$

```
(%i12) monomio:-1/6*x^3*3/4*x^2*y^3*(-2/5)*x*y^4*z;
```

(%o12) $\dfrac{x^6\,y^7\,z}{20}$

Esempio 7. Calcolare MCD e mcm fra i seguenti monomi:

a) $168a^2b^4c^3$, $1372a^3b^2$

b) $9625x^2y^3c$, $-94325x^3y^2c^5$

```
(%i1) gcd(168*a^2*b^4*c^3,1372*a^3*b^2);
```

(%o1) $28\,a^2\,b^2$

```
(%i2) gcd(9625*x^2*y^3*c,-94325*x^3*y^2*c^2);
```

(%o2) $1925\, c\, x^2\, y^2$

(%i3) `load("functs"); # serve a caricare, fra l'altro, la funzione lcm=mcm`

(%o4) /usr/share/maxima/5.9.2/share/simplification/functs.mac

(%i5) `lcm(168*a^2*b^4*c^3,1372*a^3*b^2);`

(%o5) $8232\, a^3\, b^4\, c^3$

(%i6) `lcm(9625*x^2*y^3*c,-94325*x^3*y^2*c^2);`

(%o6) $-471625\, c^2\, x^3\, y^3$

Passiamo ora alle operazioni con i polinomi. Utilizzeremo le istruzione ratsimp, expand, factor, divide, quotient e remainder.

Esempio 8. Esegui le seguenti moltiplicazioni e semplifica quando possibile i monomi simili.

a) $(a-b)(a+b)(a-2b)(a+3b)$

b) $(a^2-3b^2)(a+b)^3(a^3-8b^3)$

c) $(a-2b)(a+2b)(a+b)^2$

(%i7) `e1:(a-b)*(a+b)*(a-2*b)*(a+3*b);`

(%o7) $(a-2\,b)\,(a-b)\,(b+a)\,(3\,b+a)$

(%i8) `expand(e1);`

(%o8) $6\,b^4-a\,b^3-7\,a^2\,b^2+a^3\,b+a^4$

(%i9) `e2:(a^2-3*b^2)*(a+b)^3*(a^3-8*b^3);`

(%o9) $(b+a)^3\left(a^2-3\,b^2\right)\left(a^3-8\,b^3\right)$

(%i10) `expand(e2);`

(%o10) $24\,b^8+72\,a\,b^7+64\,a^2\,b^6-3\,a^3\,b^5-33\,a^4\,b^4-16\,a^5\,b^3+3\,a^7\,b+a^8$

(%i11) `e3:(a-2*b)*(a+2*b)*(a+b)^2;`

(%o11) $(a-2\,b)\,(b+a)^2\,(2\,b+a)$

(%i12) `expand(e3);`

(%o12) $-4\,b^4-8\,a\,b^3-3\,a^2\,b^2+2\,a^3\,b+a^4$

Esempio 9. Trovare quoziente e resto delle seguenti divisioni polinomiali.

a) $(5x^4-2x^3+x^2-3x+7)\div(x+\frac{1}{2})$

b) $(4x^6-2x^5+x^4-12x^2+42x^2-20x-69)\div(x+3)$

(%i13) `p[1]:5*x^4-2*x^3+x^2-3*x+7;`

(%o13) $5\,x^4-2\,x^3+x^2-3\,x+7$

(%i14) `p[2]:x+1/2;`

(%o14) $x+\dfrac{1}{2}$

(%i15) `quotient(p[1],p[2]);`

(%o15) $\dfrac{40\,x^3-36\,x^2+26\,x-37}{8}$

(%i16) `remainder(p[1],p[2]);`

(%o16) $\frac{149}{16}$

(%i17) `s[1]:4*x^6-2*x^5+x^4-12*x^2+42*x^2-20*x-69;`

(%o17) $4\,x^6-2\,x^5+x^4+30\,x^2-20\,x-69$

(%i18) `s[2]:x+3;`

(%o18) $x+3$

(%i19) `divide(s[1],s[2]);`

(%o19) $\left[4\,x^5-14\,x^4+43\,x^3-129\,x^2+417\,x-1271, 3744\right]$

Esempio 10. Fattorizza i seguenti polinomi.

a) $2x^4+5x^3-8x^2-17x-6$

b) $6x^6-29x^5+24x^4+40x^3-36x^2-11x+6$

(%i20) `factor(2*x^4+5*x^3-8*x^2-17*x-6);`

(%o20) $(x-2)\,(x+1)\,(x+3)\,(2\,x+1)$

(%i21) `factor(6*x^6-29*x^5+24*x^4+40*x^3-36*x^2-11*x+6);`

(%o21) $(x-3)\,(x-2)\,(x-1)\,(x+1)\,(2\,x+1)\,(3\,x-1)$

Esercizio 5. Esegui le seguenti divisioni:

a) $(x^2-x-12)\div(x-4)$

b) $(x^5+x^2-x^4-x)\div(x-1)$

Esercizio 6. Fattorizza i seguenti polinomi:

a) $2x^4+10x^3-50x^2-250x$

b) $3a^2x^3-15a^3x^2-42a^4x$

Esempio 11. Semplifica le seguenti frazioni algebriche:

$$\frac{x^3+3\,x^2-x-3}{x^2+2\,x+1}$$

$$\left(\frac{1}{a-b}+\frac{1}{a+b}\right)\cdot\frac{4a-4b}{\mathrm{ab}+b^2}\cdot\frac{a^2+2\mathrm{ab}+b^2}{4a}$$

(%i22) `Frac:(x^3+3*x^2-x-3)/(x^2+2*x+1);`

(%o22) $\frac{x^3+3\,x^2-x-3}{x^2+2\,x+1}$

(%i23) `expand(Frac);`

(%o25) $\frac{x^3}{x^2+2\,x+1}+\frac{3\,x^2}{x^2+2\,x+1}-\frac{x}{x^2+2\,x+1}-\frac{3}{x^2+2\,x+1}$

(%i26) `factor(Frac);`

(%o26) $\frac{(x-1)\,(x+3)}{x+1}$

(%i27) `ratsimp(Frac);`

(%o27) $\frac{x^2+2\,x-3}{x+1}$

Talvolta è conveniente utilizzare fullratsimp in luogo di ratsimp perché vengono eseguite delle semplificazioni “non algebriche”.

(%i1) `expr: (x^(a/2)+1)^2*(x^(a/2)-1)^2/(x^a-1);`

(%o4) $$\frac{\left(x^{\frac{a}{2}}-1\right)^{2}\left(x^{\frac{a}{2}}+1\right)^{2}}{x^{a}-1}$$

(%i5) `ratsimp(expr);`

(%o5) $$\frac{x^{2a}-2\,x^{a}+1}{x^{a}-1}$$

(%i6) `fullratsimp(expr);`

(%o6) $x^{a}-1$

Esercizio 7. Semplifica la seguente frazione algebrica:

$$\left(\frac{1}{a-b}+\frac{1}{a+b}\right)\cdot\frac{4a-4b}{\mathrm{ab}+b^{2}}\cdot\frac{a^{2}+2\mathrm{ab}+b^{2}}{4a}$$

Successivamente trova il suo valore facendo le seguenti sostituzioni: $a=3, b=2$

Sono possibili altre manipolazioni di espressioni algebriche/trascendenti. Per esempio si può dividere una espressione in un numero predefinito di polinomi, oppure è possibile prendere i primi n monomi di un polinomio dopo che è stato ordinato, oppure è possibile dividere un polinomio in due polinomi: quello che contiene una certa variabile e quello che non la contiene. Gli esempi che seguono dovrebbero essere chiarificatori.

(%i7) `partition(x^2+x*y+3*y+x-1,x); # x è la variabile che discrimina`

(%o7) $\left[3\,y-1, x\,y+x^{2}+x\right]$

(%i7) `part(x^3-x^2+x-1,2); # prendo il secondo elemento da sinistra`

(%o8) $-x^{2}$

(%i9) `part(-x^2+x-1+x^3,2); # idem dopo il riordinamento (medesimo risultato!)`

(%o9) $-x^{2}$

(%i10) `expr: (x+y)/2-sqrt(x-sqrt(x+1))+log(sin(sqrt(x+y)));`

(%o10) $\log\left(\sin\left(\sqrt{y+x}\right)\right)+\frac{y+x}{2}-\sqrt{x-\sqrt{x+1}}$

(%i11) `pickapart(expr,0);`

(%t11) $\log\left(\sin\left(\sqrt{y+x}\right)\right)+\frac{y+x}{2}-\sqrt{x-\sqrt{x+1}}$

(%o11) $\%t11$

(%i12) `pickapart(expr,1);`

(%t12) $\log\left(\sin\left(\sqrt{y+x}\right)\right)$

(%t13) $\frac{y+x}{2}$

(%t14) $-\sqrt{x-\sqrt{x+1}}$

(%o14) $\%t14+\%t13+\%t12$

(%i14) `pickapart(expr,2);`

(%t15) $\sin\left(\sqrt{y+x}\right)$

(%t16) $y+x$

(%t17) $\sqrt{x-\sqrt{x+1}}$

(%o17) $-\%t17+\frac{\%t16}{2}+\log\left(\%t15\right)$

(%i17) `pickapart(expr,3);`

(%t18) $\sqrt{y+x}$

(%t19) $x-\sqrt{x+1}$

(%o19) $\frac{y+x}{2}-\sqrt{\%t19}+\log\left(\sin\left(\%t18\right)\right)$

Se l'erpressione è complessa utilizzando realpart e imagpart la si scompone nella sua parte reale e in quella immaginaria.

(%i19) `z: (x+%i*y)^2;`

(%o19) $\left(i\,y+x\right)^2$

(%i20) `[realpart(z),imagpart(z)];`

(%o20) $\left[x^2-y^2, 2\,x\,y\right]$

4 Definire Espressioni e Funzioni

Quando si eseguono operazioni complesse, che coinvolgono espressioni lunghe e comlicate da trascrivere, è certamente utile dare loro un nome cosicché sia possibile utilizzare il nome in luogo delle espressioni. Come abbiamo visto nella sezione precedente Maxima fa questa operazione in modo automatico con ogni riga di input e di output usando %i(numero) e %o(numero). Per personalizzare i nome si utilizza la seguente forma (che di fatto abbiamo già usato!):

```
NomeEspressione : espressione;
```

L'esempio seguente, mostra questo utilizzo, si noti che inserendo semplicemente il nome della espressione si ottiene come risultato l'espressione.

(%i14) `x:102/50+17/11;`

(%o14) $\frac{986}{275}$

(%i15) `y:sqrt(x);`

(%o15) $\frac{\sqrt{986}}{5\,\sqrt{11}}$

(%i16) `y;`

(%o16) $\frac{\sqrt{986}}{5\,\sqrt{11}}$

In matematica, scienze, ingegneria e fisica le lettere greche vengono utilizzate con profusione. Maxima riconosce nomi[1] come alpha, beta, rho, come viene mostrato nell'esempio. Si noti l'uso delle parentesi quadrate per denotare delle variabili indicizzate come in beta[rho].

(%i9) `beta[z] : sqrt(beta^2-beta[rho]^2);`

(%o9) $\sqrt{\beta^2 - \beta_\rho^2}$

(%i10) `beta[z];`

(%o10) $\sqrt{\beta^2 - \beta_\rho^2}$

(%i11) `beta[rho];`

(%o11) β_ρ

In Maxima le funzioni si definiscono come segue:

```
f(x) := espressione;
```

L'uso viene mostrato qui di seguito. Si noti che le espressioni possono contenere anche variabili indefinite, quando però a queste si assegna un valore, la funzione riflette questo variazione.

(%i1) `f : exp(-sin(x^2));`

(%o1) $\mathrm{e}^{-\sin\left(x^2\right)}$

(%i2) `plot2d(f,[x,0,5]);`



5 Risoluzione di Equazioni

Il comando solve risolve, quando possibile, una o più equazioni (sistemi). Negli esempi che seguiranno calcoleremo le soluzioni di equazioni e sistemi.

Esempio 12. Risolvere l'equazione:

$$8\left(\frac{1}{2} - x\right) - 3(2x-1) = 2(x+3) - 11$$

(%i28) `solve(8*(1/2-x)-3*(2*x-1)=2*(x+3)-11);`

1. Si noti che sono scritti in inglese. Altri: epsilon, phi, psi, lambda, ...

(%o28) $\left[x=\frac{3}{4}\right]$

Esempio 13. Risolvere la seguente equazione parametrica rispetto alla variabile x.

$$x(a-3)+\frac{2(x+1)}{a}=1$$

(%i29) `eq:x*(a-3)+2*(x+1)/a=1;`

(%o29) $\frac{2\,(x+1)}{a}+(a-3)\,x=1$

(%i30) `ratsimp(eq);`

(%o31) $\frac{\left(a^2-3\,a+2\right)x+2}{a}=1$

(%i32) `factor(a^2-3*a+2);`

(%o34) $(a-2)\,(a-1)$

(%i35) `solve(eq,x);`

(%o35) $\left[x=\frac{1}{a-1}\right]$

Esercizio 8. Risolvere le seguenti equazioni intere:

a) $3(2x+1)-2(3x+1)=4x+3-4(x-1)$

b) $\frac{x-4}{5}-\frac{2-x}{4}+\frac{7}{10}\left(\frac{3x-1}{2}+\frac{2(x-3)}{5}\right)=0$

c) $a(x-1)+\mathrm{ax}=x(2a+1)+1+3a$

Vediamo ora un esempio nel quale viene risolto un sistema lineare di due equazioni in due incognite. Useremo il comando solve i cui argomenti saranno due liste: la lista delle equazioni e la lista delle incognite. Le lista sono racchiuse da parentesi quadrate [...] e gli elementi sono separati dalla virgola.

Esempio 14. Risolvi il seguente sistema lineare:

$$\begin{cases} 4x-5y=3 \\ 2x+3y=1 \end{cases}$$

(%i36) `solve([4*x-5*y=3,2*x+3*y=1],[x,y]);`

(%o36) $\left[\left[x=\frac{7}{11}, y=-\frac{1}{11}\right]\right]$

In questo modo risolviamo senza conoscere il metodo che Maxima applica. Supponiamo di voler applicare il metodo di Cramer, in questo caso dobbiamo scrivere la matrice dei coefficiente e successivamente mettere il sistema in forma matriciale.

(%i37) `A:matrix([4,-5],[2,3]);`

(%o3) $\begin{pmatrix} 4 & -5 \\ 2 & 3 \end{pmatrix}$

(%i4) `X:matrix([x[1]],[x[2]]);`

(%o4) $\begin{pmatrix} x_1 \\ x_2 \end{pmatrix}$

(%i5) `B:matrix([3],[1]);`

(%o5) $\begin{pmatrix} 3 \\ 1 \end{pmatrix}$

(%i6) A.X=B;

(%o6) $\begin{pmatrix} 4\,x_1 - 5\,x_2 \\ 3\,x_2 + 2\,x_1 \end{pmatrix} = \begin{pmatrix} 3 \\ 1 \end{pmatrix}$

(%i7) X=invert(A).B;

(%o7) $\begin{pmatrix} x_1 \\ x_2 \end{pmatrix} = \begin{pmatrix} \frac{7}{11} \\ -\frac{1}{11} \end{pmatrix}$

E' possibile utilizzare anche una istruzione ottimizzata per sistemi lineari, come viene mostrato nell'esempio.

(%i8) linsolve([4*x-5*y=3,2*x+3*y=1],[x,y]);

(%o8) $\left[x = \frac{7}{11}, y = -\frac{1}{11}\right]$

Esercizio 9. Risolvere i seguenti sistemi lineari:

a) $\begin{cases} 3(x+y) - 6y + 8 - 3x = 2(3x - 2y) \\ 2x + 3(2y - 3x) = 2x + 1 \end{cases}$

b) $\begin{cases} (x+1)^2 - 2y = x^2 - 2x + 3 \\ (x-2)(x+3) + 3x = y + (x-2)^2 \end{cases}$

Nell'esempio viene risolta l'equazione $e^{x^2-3x+2} = 1/2$ rispetto alla variabile x, la riga di output successiva mostra la soluzione esatta. L'uso di float restituisce i valori numerici ed evidenzia che le soluzioni sono numeri complessi.

In Maxima la funzione log è il logaritmo naturale che in altri ambiti viene indicato con $\ln(x)$ oppure con $\log_e(x)$. Si noti che nonostante l'equazione sia stata nominata con la variabile eq, si può sempre utilizzare %o2, avremmo potuto utilizzare anche il comando solve(eq,x);.

(%i2) eq : exp(x^2-3*x+2)=1/2;

(%o2) $\mathrm{e}^{x^2-3x+2} = \frac{1}{2}$

(%i3) solve(%o2,x);

(%o3) $\left[x = -\frac{\sqrt{1-4\log(2)}-3}{2}, x = \frac{\sqrt{1-4\log(2)}+3}{2}\right]$

(%i4) float(%o3);

(%o4) $[x = -0.5\,(1.33138601548904\,\mathrm{i} - 3.0), x = 0.5\,(1.33138601548904\,\mathrm{i} + 3.0)]$

Una istruzione ottimizzata per la risoluzione di sistemi algebrici è algsys ha che sintassi analoga a solve e che utlizzeremo più avanti.

6 Limiti

Il calcolo di limiti è molto semplice, basta utilizzare la funzione limit(f,var,valore[,verso]) dove f rappresenta la funzione, var la variabile rispetto alla quale si vuole calcolare, valore è il valore sul quale si vuole calcolare il limite e, opzionalmente, si può specificare il verso plus o minus per indicare il limite destro e sinistro.

Esempio 15. Calcolare i seguenti limiti:

$$\lim_{x \to 0} \frac{\sin(x)}{x}$$

$$\lim_{x \to 0^-} \frac{x}{|x|}$$

$$\lim_{x \to +\infty} \left(1 + \frac{1}{x}\right)^x$$

(%i6) `limit(sin(x)/x,x,0);`

(%o6) 1

(%i7) `limit(x/abs(x),x,0);`

(%o7) $\lim_{x \to 0} \frac{x}{|x|}$

(%i8) `limit(x/abs(x),x,0,minus);`

(%o8) -1

(%i9) `assume(x>0);`

(%o9) $[x > 0]$

(%i10) `limit(x/abs(x),x,0);`

(%o10) 1

(%i11) `limit((1+1/x)^x,x,inf);`

(%o11) e

Nel secondo limite si vede come calcolare il limite destro o sinistro, in alternativa si può utilizzare il comando assume. Si noti che se non ci sono le condizioni, il limite non viene calcolato. Il risultato del terzo limite è la costante di Nepero e.

E' possibile combinare anche funzionali diversi, per esempio possiamo calcolare un integrale generalizzato.

Esempio 16. Si calcoli il seguente integrale generalizzato:

$$\int_0^{+\infty} e^{-t^2} \mathrm{dt}$$

(%i23) `f(x):=integrate(%e^(-t^2),t,0,x);`

(%o23) $f(x) := \mathrm{integrate}\left(\mathrm{e}^{-t^2}, t, 0, x\right)$

(%i24) `limit(f(x),x,+inf);`

(%o26) $\frac{\sqrt{\pi}}{2}$

7 Derivate

L'operatore diff calcola la derivata simbolica della funzione specificata come primo argomento rispetto alla variabile del secondo.

(%i17) `g:exp(sin(t^2));`

(%o24) $e^{\sin(t^2)}$

```
(%i25) diff(g,t);
```

(%o26) $2\,t\cos\left(t^2\right)e^{\sin(t^2)}$

Opzionale è il terzo argomento che indica l'ordine della derivata, il valore predefinito è uno (viene calcolata la derivata prima).

```
(%i28) diff(g,t,2);
```

(%o28) $-4\,t^2\sin\left(t^2\right)e^{\sin(t^2)}+4\,t^2\cos\left(t^2\right)^2e^{\sin(t^2)}+2\cos\left(t^2\right)e^{\sin(t^2)}$

E' anche possibile verificare se una data funzione è soluzione di una certa equazione differenziale, come mostra il seguente esempio:

```
(%i30) y:exp(-t^2);
```

(%o30) e^{-t^2}

```
(%i31) diff(y,t,2)+2*t*diff(y,t)+2*y;
```

(%o32) 0

Che ci permette di concludere che la funzione e^{-t^2} è soluzione della equazione differenziale:

$$y''+2\mathrm{xy}'+2y=0$$

Si può utilizzare diff anche annidato tipicamente durante il calcolo di derivate parziali. Nell'esempio che segue calcoliamo $\frac{\partial^2}{\partial t\partial s}\sin(\beta_t t)\cos(\beta_s s)e^{-i\beta_w w}$.

```
(%i33) f:sin(beta[t])*t*cos(beta[s]*s)*exp(-%i*beta[w]*w);
```

(%o34) $\cos\left(s\,\beta_s\right)t\sin\left(\beta_t\right)e^{-iw\beta_w}$

```
(%i35) diff(diff(f,t),s);
```

(%o35) $-\beta_s\sin\left(s\,\beta_s\right)\sin\left(\beta_t\right)e^{-iw\beta_w}$

Esercizio 10. Calcolare la derivata delle seguenti funzioni:

$$y=e^{\sin(x)}\cos(1/x)$$

$$Y=\left(1+\frac{1}{x}\right)^{\frac{1}{x}}$$

$$y=\frac{\sin(x^2)+\cos(x^2)}{\tan(x^2)}$$

Se vogliamo istruire Maxima sulla dipendenza di una funzione da certe variabili possiamo utilizzare il comando depends(funzione,variabili) questa possibilità è molto utile nel caso si vogliano verificare delle formule. L'istruzione remove(funzione, dependency) rimuove tutte le dipendenze della funzione f.

```
(%i1) diff(f,x);
```

(%o1) 0

```
(%i2) depends([f,g],x)$
(%i3) diff(f*g,x);
```

(%o3) $f\left(\frac{d}{d\,x}\,g\right)+\frac{d}{d\,x}\,f\,g$

```
(%i4) remove([f,g],dependency)$
(%i5) diff(f*g,x);
```

(%o5) 0

8 Integrali

Il calcolo degli integrali indefiniti (le primitive) può essere eseguito utilizzando l'istruzione integrate. Il primo argomento è la funziona della quale si vuole calcolare la primitiva, il secondo argomento è la variabile rispetto alla quale fare il calcolo (la variabile di integrazione). Nell'esempio seguente calcoliamo delle primitive sia direttamente che attraverso la definizione preventiva delle funzioni integrande.

(%i37) `integrate(t^2*sin(t),t);`

(%o40) $2\,t\sin\left(t\right)+\left(2-t^2\right)\cos\left(t\right)$

(%i41) `f:t^2*sin(t);`

(%o41) $t^2\sin\left(t\right)$

(%i42) `integrate(f,t);`

(%o42) $2\,t\sin\left(t\right)+\left(2-t^2\right)\cos\left(t\right)$

(%i43) `integrate(exp(t^3),t);`

(%o44) $\int e^{t^3}\,d\,t$

E' possibile anche calcolare integrali definiti, basta specificare come terzo e quarto argomento gli estremi di integrazione.

(%i45) `integrate(f,t,0,%pi);`

(%o45) π^2-4

Prcedendo il comando integrate da un apostrofo[2] si fa in modo che Maxima non esegua il calcolo.

(%i47) `'integrate(f,t,0,5*%pi);`

(%o47) $\int_0^{5\pi} t^2\sin\left(t\right)\,d\,t$

E' anche possibile cambiare la variabile con il comando changevar come si evidenzia dall'esempio seguente.

(%i48) `changevar(%o47,t^2-s,s,t);`

(%o54) $\dfrac{\int_0^{25\pi^2}\sin\left(\sqrt{s}\right)\sqrt{s}\,d\,s}{2}$

Il primo argomento rappresenta l'integrale, il secondo il cambiamento di variabile, cioè la equazione che lega le due variabili (si può scrivere $s=t^2$ oppure t^2-s), il terzo la nuova variabile e il quarto la vecchia variabile.

Si possono calcolare anche integrali multipli, basta annidare la funzione integrate. Nell'esempio seguente calcoleremo $\int_0^1 \int_0^y e^{-t}\mathrm{d}t\ \mathrm{d}y$. Si noterà che Maxima chiede se y è positivo, negativo o nullo, noi risponderemo positive; (non dimentiare il punto e virgola finale).

(%i57) `integrate(integrate(exp(-s),s,0,w),w,0,1);`

Is w positive, negative, or zero?`positive;`

(%o59) $e^{-1}\left(e+1\right)-1$

2. In verità ciò succede anche con tutti gli altri operatori.

```
(%i60) ratsimp(%);
```

(%o60) e^{-1}

```
(%i61) float(%);
```

(%o61) 0.36787944117144

Volendo potremmo definire una funzione utilizzando l'integrale $F(t) = \int_0^t \exp(-s)\mathrm{ds}$, e verificare il teorema fondamentale del calcolo integrale[3].

```
(%i62) F(t):='integrate(exp(-s),s,0,t);
```

(%o18) $F(t) := \int_0^t \exp(-s)\,\mathrm{d}\,s$

```
(%i19) diff(F(t),t);
```

Is t positive, negative, or zero?`positive;`

(%o68) e^{-t}

Se vogliamo calcolare un integrale definito approssimato, possiamo utilizzare il comando romberg[4] con la stessa sintassi di integrate.

```
(%i69) romberg(exp(-t^3),t,0,1);
```

(%o69) 0.80751119008366

```
(%i70) f(t):=exp(-t^3)*sin(t);
```

(%o70) $f(t) := \exp(-t^3)\sin(t)$

```
(%i71) romberg(f(t),t,0,1);
```

(%o71) 0.32619406816191

Esercizio 11. Calcolare i seguenti integrali:

$$\int \sin(x)\log(x)\mathrm{dx}$$

$$\int \frac{x^2 - x + 1}{x^4 + 2x^2 + 1}\mathrm{dx}$$

$$\int_0^1 \sin(\cos(x))\mathrm{dx}$$

9 Vettori e Matrici

Generalmente, per calcoli con matrici e vettori numerici è preferibile utilizzare strumenti come Scilab o Octave. Maxima è consigliato nel casi si debbano fare calcoli simbolici o quando si vogliono otterere risultati esatti.

I comandi matrix e entermatrix(n,m)[5] permettono di dichiarare una matrice di arbitraria dimensione, se si vuole costruire la matrice identità si usa il comando ident(dim) mettendo al posto di dim la dimensione. Per costruire nuove matrici si possono utilizzare i comandi addcol e addrow che affiancano o sovrappongono matrici fra loro.

```
(%i21) A: matrix([a,b],[c,d]);
```

3. Si noti l'uso dell'apice (') prima della istruzione integrate che impedisce a Maxima di eseguire l'istruzione.

4. Che, ovviamente, utilizza il metodo di Romberg. Sono previsti anche altri metodi di integrazione numerica per questo si fa riferimento al manuale di Maxima.

5. Il comando è molto utile per i principianti perché li guida sia nella scelta della tipologia della matrice sia nell'inserimento degli elementi, i seguito apprezzerete la più scarna istruzione matrix.

(%o21) $\begin{pmatrix} a & b \\ c & d \end{pmatrix}$

```
(%i22) B: addcol(A,A);
```

(%o22) $\begin{pmatrix} a & b & a & b \\ c & d & c & d \end{pmatrix}$

```
(%i23) C: addrow(A,A);
```

(%o23) $\begin{pmatrix} a & b \\ c & d \\ a & b \\ c & d \end{pmatrix}$

Le limitazioni dei comandi addrow e addcol sono evidenti: posso sovrapporre due matrici se hanno lo stesso numero di colonne, mentre le posso affiancare se hanno lo stesso numero di righe.

Se, viceversa, si vogliono estrarre delle righe o delle colonne da una matrice ai usano i comandi col e row, come mostra il seguente esempio.

```
(%i24) A: matrix([0,1,a],[1,b,0],[c,0,0]);
```

(%o24) $\begin{pmatrix} 0 & 1 & a \\ 1 & b & 0 \\ c & 0 & 0 \end{pmatrix}$

```
(%i25) row(A,2);
```

(%o25) $\begin{pmatrix} 1 & b & 0 \end{pmatrix}$

```
(%i26) col(A,3);
```

(%o27) $\begin{pmatrix} a \\ 0 \\ 0 \end{pmatrix}$

```
(%i28) I: ident(3);
```

(%o38) $\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$

Determinanti e inverse di matrici possono essere calcolati con i comandi determinant e invert mentre il comando adjoint restituisce la matrice aggiunta (cioè la matrice inversa moltiplicata per il determinante).

```
(%i28) determinant(A);
```

(%o28) $-a\,b\,c$

```
(%i29) invert(A);
```

(%o29) $\begin{pmatrix} 0 & 0 & \frac{1}{c} \\ 0 & \frac{1}{b} & -\frac{1}{b\,c} \\ \frac{1}{a} & -\frac{1}{a\,b} & \frac{1}{a\,b\,c} \end{pmatrix}$

```
(%i30) adjoint(A);
```

(%o30) $\begin{pmatrix} 0 & 0 & -a\,b \\ 0 & -a\,c & a \\ -b\,c & c & -1 \end{pmatrix}$

```
(%i31) invert(A)*determinant(A);
```

$$(\%o31)\quad \begin{pmatrix} 0 & 0 & -a\,b \\ 0 & -a\,c & a \\ -b\,c & c & -1 \end{pmatrix}$$

L'operazione di moltiplicazione (riga/colonna) si esegue col comando . mentre per la somma e la sottrazione si usano i soliti simboli.

```
(%i32) B: invert(A);
```

$$(\%o32)\quad \begin{pmatrix} 0 & 0 & \frac{1}{c} \\ 0 & \frac{1}{b} & -\frac{1}{b\,c} \\ \frac{1}{a} & -\frac{1}{a\,b} & \frac{1}{a\,b\,c} \end{pmatrix}$$

```
(%i33) A.B;
```

$$(\%o33)\quad \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Attenzione: se si usa il simbolo * fra due matrici o si eleva una matrice a un certo esponente, si agisce su ogni singolo elemento.

```
(%i34) C: matrix([1,2],[2,1]);
```

$$(\%o34)\quad \begin{pmatrix} 1 & 2 \\ 2 & 1 \end{pmatrix}$$

```
(%i35) C*2;
```

$$(\%o35)\quad \begin{pmatrix} 2 & 4 \\ 4 & 2 \end{pmatrix}$$

```
(%i36) C*C;
```

$$(\%o36)\quad \begin{pmatrix} 1 & 4 \\ 4 & 1 \end{pmatrix}$$

```
(%i37) C^(-1);
```

$$(\%o37)\quad \begin{pmatrix} 1 & \frac{1}{2} \\ \frac{1}{2} & 1 \end{pmatrix}$$

Il polinomio caratteristico di una matrice si ottiene calcolando direttamente $\boldsymbol{p(\alpha) = \det(A - \alpha I)}$), oppure con il comando charpoly.

```
(%i39) expand(determinant(A-alpha*ident(3)));
```

$$(\%o41)\quad -a\,b\,c + a\,\alpha\,c + \alpha^2\,b - \alpha^3 + \alpha$$

```
(%i42) expand(charpoly(A,alpha));
```

$$(\%o44)\quad -a\,b\,c + a\,\alpha\,c + \alpha^2\,b - \alpha^3 + \alpha$$

Il comando eigenvalue di una matrice quadrata restituisce gli autovalori sotto forma di due liste, la prima contiene gli autovalori mentre la seconda le loro molteplicità. Il comando eigenvectors fornisce per ogni autovalore i corrispondenti gli autovettori.

```
(%i45) A: matrix([0,1,0,0],[3*w^2,0,0,2*w],[0,0,0,1],[0,-2*w,0,0]);
```

(%o45) $\begin{pmatrix} 0 & 1 & 0 & 0 \\ 3\,w^2 & 0 & 0 & 2\,w \\ 0 & 0 & 0 & 1 \\ 0 & -2\,w & 0 & 0 \end{pmatrix}$

```
(%i46) eigenvalues(A);
```

(%o46) $[[-\mathrm{i}\,w, \mathrm{i}\,w, 0], [1, 1, 2]]$

```
(%i47) eigenvectors(A);
```

(%o47) $[[[-\mathrm{i}\,w, \mathrm{i}\,w, 0], [1, 1, 2]], [1, -\mathrm{i}\,w, -2\,\mathrm{i}, -2\,w], [1, \mathrm{i}\,w, 2\,\mathrm{i}, -2\,w], [0, 0, 1, 0]]$

Esercizio 12. Risolvere il seguente sistema lineare mettendolo in forma matriciale.

$$\begin{cases} x - y + z = 1 \\ x + y + 2z = 2 \\ -3x + y - z = 1 \end{cases}$$

Esercizio 13. Data la matrice:

$$\mathrm{A} = \begin{pmatrix} 2 & 1 & 1 \\ 1 & 2 & 1 \\ 1 & 1 & 2 \end{pmatrix}$$

calcolare il polinomio caratteristico $p(\gamma)$ e verificare che la matrice $p(A)$ è la matrice nulla.

10 Disegni in 2D e 3D

Maxima utilizza il programma gnuplot per rappresentare graficamente dei grafici. Di seguito vedremo la sintassi e l'uso dei due comandi più importanti per il disegno: plot2d e plot3d.

Il comando plot2d può essere molto utile per localizzare le radici di una equazione e per successivamente calcolarle con un metodo numerico, p.e. quello di Newton.

Esempio 17. Determinare le radici della seguente equazione: $\sin(x) - \log(x) = 0$. Evidentemente l'equazione non è risolvibile per via elementare. Facciamo il grafico della funzione $y = \sin(x) - \log(x)$ evidenziando gli assi cartesiani.

```
(%i48) y: sin(x)-log(x)$
(%i49) plot2d(y,[x,0,5],[gnuplot_preamble, "set zeroaxis"])$
(%i52)
```



Evidentemente la soluzione è "vicina" a 2.5, perciò useremo il metodo di Newton partenda da 2.5.

```
(%i52) load("newton")$
(%i53) newton(y,2.5);
```

```
Warning:  Float to bigfloat conversion of 2.5
```
(%o53) $2.219107150437273B0$

Esempio 18. Risolvere numericamente e graficamente il sistema:

$$\begin{cases} y = 2x - 3 \\ y = -x + 6 \end{cases}$$

```
(%i54) linsolve([alpha=2*beta-3, alpha=-beta+6],[alpha,beta]);
```
(%o58) $[\alpha = 3, \beta = 3]$

```
(%i59) plot2d([2*beta-3,-beta+6],[beta,0,5],[gnuplot_preamble, "set zeroaxis"])$
```



Evidentemente[6] la coppia delle soluzioni è circa $(3, 3)$.

Maxima offre anche la possibilità di tracciare dei grafici in tre dimensioni, nota l'equazione cartesiana della superficie $\gamma = f(\alpha, \beta)$.

Esempio 19. Disegnare il grafico della funzione $\gamma = \cos(\mathrm{sqrt}(\alpha^2 + \beta))$

```
(%i60) plot3d(cos(sqrt(alpha^2+beta)),[alpha,-5,5],[beta,-5,5],
       [plot_format,gnuplot], [gnuplot_pm3d,true]);
```



Esempio 20. Verificare se nell'intervallo $[2, 4]$ è applicabile il teorema di Rolle alla funzione:

$$f(x) = \begin{cases} -3x + 6 \text{ per } x < 3 \\ x^2 - 6x + 8 \text{ per } x \geqslant 3 \end{cases}$$

6. Del senno del poi son piene le fosse! :-))

Innanzitutto la funzione è certamente continua e derivabile $\forall x \neq 3$ essendo composizione di funzioni continue. Indaghiamo sulla continuità in 3 e sul valore assunto agli estremi dell'intervallo. Utilizzeremo oltre ai comandi noti anche l'istruzione block della quale discuteremo nella sezione riguardante la programmazione con Maxima.

```
(%i1) f1(x):=-3*x+6$
(%i2) f2(x):=x^2-6*x+8$
(%i3) f(x):=block([], if x<3 then return(f1(x)) else return(f2(x)))$
(%i4) is(f(2)=f(4));
(%o4) true
(%i5) limit(f1(x),x,3,minus);
(%o6) -3
(%i7) limit(f2(x),x,3,plus);
(%o7) -1
(%i8) plot2d(f,[x,2,4])$
```

Come appare dal grafico 3 è un punto di discontinuità perciò il teorema non è applicabile.



Esempio 21. Trovare una approssimazione di $\cos(x)$ al secondo ordine intorno a zero e visualizzarne la bontà in intorni via via più grandi.

Utilizzeremo il comando Taylor(f,x,0,2) perché ci interessa lo sviluppo fino al secondo ordine.

(%i62) `taylor(cos(x),x,0,2);`

(%o62) $1 - \frac{x^2}{2} + \cdots$

(%i63) `trunc(%); # Adesso diventa un polinomio`

(%o63) $1 - \frac{x^2}{2} + \cdots$

```
(%i64) f:%$
(%i70) errore(x):= (cos(x)-f)$
(%i88) x:0.01$
(%i89) errore(x);
```

(%o89) 0.99995000041667 –

```
(%i90) x:0.1$
(%i91) errore(x);
```

(%o91) 0.99500416527803 –

```
(%i92) x:0.2
```

(%o96) 0.2

```
(%i97) errore(x);
```

(%o98) 0.98006657784124 –

```
(%i99) remvalue(x)$
(%i105) plot2d(cos(x)-(1-x^2/2),[x,-%pi/2,%pi/2],[gnuplot_preamble, "set
        zeroaxis"])$
```



Concludendo fino a un angolo di circa 0.2 rad (cioè circa 12^o) l'errore è circa del 2%.

Esercizio 14. Risolvere numericamente le seguenti equazioni:

- $\log(x^2+1) = \cos(x)$
- $e^{-x^2} = x^4 - 3$
- $e^{\sin(x)} = x + \frac{6}{5}$

11 Programmare

(da fare)

12 Applicazioni

12.1 Urti elastici

Un carrello di massa m_1 e velocità v_1^i urta elasticamente un carrello di massa m_2 inizialmente fermo. Determinare le velocità dei carrelli dopo l'urto e discuterle supponendo $m_1 = m_2$, $m_1 \ll m_2$ e $m_1 \gg m_2$.

Siamo nella situazione di poter applicare la conservazione della quantità di moto e la conservazione dell'energia cinetica[7], indicando con v_{1f} e v_{2f} rispettivamente le velocità dopo l'urto del primo e del secondo carrello, possiamo scrivere:

$m_1 v_{1i} = m_1 v_{1f} + m_2 v_{2f}$

$\frac{1}{2} m_1 v_{1i}^2 = \frac{1}{2} m_1 v_{1f}^2 + \frac{1}{2} m_2 v_{2f}^2$

Risolviamo le equazioni:

```
(%i1) eq1:m1*v1i=m1*v1f+m2*v2f$
(%i5) eq2:1/2*m1*v1i^2=1/2*m1*v1f^2+1/2*m2*v2f^2$
(%i6) algsys([eq1,eq2],[v1f,v2f]);
```

(%o6) $$\left[[\mathrm{v1f} = \mathrm{v1i}, \mathrm{v2f} = 0], \left[\mathrm{v1f} = -\frac{(\mathrm{m2} - \mathrm{m1})\,\mathrm{v1i}}{\mathrm{m2} + \mathrm{m1}}, \mathrm{v2f} = \frac{2\,\mathrm{m1}\,\mathrm{v1i}}{\mathrm{m2} + \mathrm{m1}}\right]\right]$$

```
(%i7) v1:%o6[2][1]
```

(%o7) $$\mathrm{v1f} = -\frac{(\mathrm{m2} - \mathrm{m1})\,\mathrm{v1i}}{\mathrm{m2} + \mathrm{m1}}$$

```
(%i8) v2:%o6[2][2]
```

(%o8) $$\mathrm{v2f} = \frac{2\,\mathrm{m1}\,\mathrm{v1i}}{\mathrm{m2} + \mathrm{m1}}$$

Caso 1: $m_1 = m_2$.

```
(%i9) subst(m1,m2,v1)
```

(%o9) $\mathrm{v1f} = 0$

```
(%i10) subst(m1,m2,v2)
```

(%o10) $\mathrm{v2f} = \mathrm{v1i}$

Dunque dopo l'urto il primo carrello si ferma e il secondo procede con la stessa velocità del primo (plausibile).

Caso 2: $m_1 \ll m_2$. Per considerare questa situazione basterà calcolare il limite per m_2 che tende a più infinito.

```
(%i11) limit([v1,v2],m2,inf)
```

Is m1 v1i positive, negative, or zero?`positive`

(%o11) $[\mathrm{v1f} = -\mathrm{v1i}, \mathrm{v2f} = 0]$

In questa situazione m_2 è come un muro, non si muove e il primo carrello rimbalza (plausibile).

Caso 3: $m_1 \gg m_2$. Per considerare questa situazione basterà calcolare il limite per m_1 che tende a più infinito.

```
(%i12) limit([v1,v2],m1,inf)
```

(%o12) $[\mathrm{v1f} = \mathrm{v1i}, \mathrm{v2f} = 2\,\mathrm{v1i}]$

Quindi il primo carrello procede come se non fosse successo nulla (plausibile), mentre il secondo inizia a muoversi nella stessa direzione con velocità doppia (per niente plausibile).

Esercizio 15. Considerare due palloni di massa m_1 e m_2 posti a contatto uno sopra l'altro e fatti cadere per terra. Supponendo che tutti gli urti siano elastici, detrminare le velocità dopo il rimbalzo e analizzare i casi $m_1 = m_2$, $m_1 \ll m_2$ e $m_1 \gg m_2$. Supporre il pavimento una massa illimitata...

7. Dato che i carrelli si muovono orizzontalmente, l'energia potenziale gravitazionale è costante e dato che l'energia meccanica si conserva segue l'asserto.

12.2 Urti anelastici

Un carrello di massa m_1 e velocità v_{1i} urta un carrello di massa m_2 inizialmente fermo e vi resta attaccato. Determinare la velocità dei carrelli dopo l'urto e discutere la variazione della energia.

In virtù della conservazione della quantità di moto e del tipo di urto, possiamo scrivere:
$m_1 v_{1i} = m_1 v_{1f} + m_2 v_{2f}$
$v_{1f} = v_{2f}$

```
(%i1) eq1:m1*v1i=m1*v1f+m2*v2f$
(%i3) eq2:v1f=v2f$
(%i4) algsys([eq1,eq2],[v1f,v2f])
```

(%o5) $\left[\left[\mathrm{v1f} = \frac{\mathrm{m1\ v1i}}{\mathrm{m2}+\mathrm{m1}}, \mathrm{v2f} = \frac{\mathrm{m1\ v1i}}{\mathrm{m2}+\mathrm{m1}}\right]\right]$

```
(%i6) part(%o5[1][1],2)
```

(%o30) $\frac{\mathrm{m1\ v1i}}{\mathrm{m2}+\mathrm{m1}}$

```
(%i31) vf:%$
```

Calcoliamo le energie cinetiche prima e dopo l'urto.

```
(%i13) Ki:1/2*m1*v1i^2
```

(%o33) $\frac{\mathrm{m1\ v1i}^2}{2}$

```
(%i34) Kf:1/2*(m1+m2)*vf^2
```

(%o35) $\frac{\mathrm{m1}^2\,\mathrm{v1i}^2}{2\,(\mathrm{m2}+\mathrm{m1})}$

```
(%i36) Kf/Ki
```

(%o36) $\frac{\mathrm{m1}}{\mathrm{m2}+\mathrm{m1}}$

Possiamo concludere che c'è sempre perdita di energie e che è tanto maggiore quanto m_2 è più grande di m_1, mentre è piccola se m_1 è molto più grande di m_2.

12.3 Pallone gonfiato

Un pallone da basket ha la forma di una sfera. Quando ha inizialmente ha il volume di $10\,l$ viene gonfiato con una pompa che trasferisce $3\,l$ al minuto. Calcolare con quale velocità aumenta il suo raggio.

È noto che il volume di una sfera V è dato dalla formula:

$$V = \frac{4}{3}\pi r^3$$

la variabile "libera" è il tempo t, perciò deriviamo V rispetto al tempo:

```
(%i1) V(t):=4/3 * %pi * r(t)^3;
```

(%o1) $V(t) := \frac{4}{3}\pi\, r(t)^3$

```
(%i2) diff(V(t),t);
```

(%o2) $4\,\pi\, r(t)^2 \left(\frac{d}{d\,t} r(t)\right)$

Ora, dato che per $V=3$ si ricava che $r=\sqrt[3]{\frac{3}{4}\frac{V}{\pi}}=\sqrt[3]{\frac{3}{4}\frac{10}{\pi}}=\sqrt[3]{\frac{15}{2\pi}}$. Perciò basta risolvere l'equazione:

$$10=4\pi\left(\sqrt[3]{\frac{15}{2\pi}}\right)^2 x$$

(%i3) `solve(10=4*%pi*(15/(2*%pi))^(2/3)*x,x);`

(%o3) $\left[x=\frac{5\,2^{\frac{2}{3}}}{2\,15^{\frac{2}{3}}\,\pi^{\frac{1}{3}}}\right]$

(%i4) `float(%o3);`

(%o5) $[x=0.44550153919066]$

Concludendo il fattore di crescita del raggio è circa $x=0.44$.

12.4 Studio di funzione

Studiare la seguente funzione algebrica razionale:

$$f(x)=\frac{x^2-1}{x^2+1}.$$

(%i1) `f(x):=(x^2-1)/(x^2+1);`

(%o1) $f(x):=\frac{x^2-1}{x^2+1}$

(%i2) `solve(f(x)=f(-x),x); # Verifico se è pari (sì!)`

(%o2) all

(%i3) `denom(f(x)); # Isolo il denominatore`

(%o3) x^2+1

(%i4) `solve(%); # Dominio: tutti i reali`

(%o4) $[x=-\mathrm{i},x=\mathrm{i}]$

(%i5) `limit(f(x),x,inf); # Limiti`

(%o5) 1

(%i6) `diff(f(x),x,1); # Derivata prima`

(%o6) $\frac{2\,x}{x^2+1}-\frac{2\,x\,(x^2-1)}{(x^2+1)^2}$

(%i7) `ratsimp(%o6); # Semplifico la derivata prima`

(%o10) $\frac{4\,x}{x^4+2\,x^2+1}$

(%i11) `solve(denom(%o10),x); # Dominio della derivata prima: ogni reale`

(%o12) $[x=-\mathrm{i},x=\mathrm{i}]$

(%i13) `is(denom(%06)>0); # Segno del denominatore della derivata`

(%o13) **true**

(%i14) `solve(num(%o10)=0,x); # Zero della derivata`

(%o23) $[x=0]$

(%i24) `diff(f(x),x,2);`

(%o24) $$\frac{2}{x^2+1}-\frac{8\,x^2}{(x^2+1)^2}-\frac{2\,(x^2-1)}{(x^2+1)^2}+\frac{8\,x^2\,(x^2-1)}{(x^2+1)^3}$$

(%i25) `ratsimp(%); # Derivata seconda semplificata`

(%o25) $$-\frac{12\,x^2-4}{x^6+3\,x^4+3\,x^2+1}$$

(%i26) `is(denom(%)>0); # Vedi sopra`

(%o26) **true**

(%i27) `solve(num(%o25)=0,x); # Zeri della derivata seconda`

(%o27) $$\left[x=-\frac{1}{\sqrt{3}},x=\frac{1}{\sqrt{3}}\right]$$

Concludendo:

- la funzione è pari (%o2),
- ha per dominio tutti i numeri reali (%o3-4),
- ha un asintoto orizzontale $y=1$ (%o5),
- la derivata prima è positiva per $x>0$ perciò ivi cresce la funzione (%o5-23),
- la derivata seconda è positiva nell'intervallo: $[-\frac{1}{\sqrt{3}},+\frac{1}{\sqrt{3}}]$, perciò ivi ha concavità verso l'alto (%o24-27).

Ora il grafico che ottengo con il comando
plot2d(f(x),[x,-10,10],[gnuplot_preamble, "set zeroaxis;"]);



12.5 Sommatorie

Questionario n. 2. Sessione ordinaria e supplettiva dell'Esame di Stato 2003-2004.
Determinare il più grande valore di n per cui l'espressione numerica $\sum_{k=5}^{n} k$ non supera 10000.

I calcoli sono piuttosto facili, basta ricordate che $\sum_{k=1}^{n} k = \frac{n(n+1)}{n}$ e togliere 10 che è la somma dei primi quattro numeri ecc., ma con Maxima è ancora più facile.

(%i1) `sum(k,k,5,n), simpsum;`

(%o1) $\frac{n^2+n}{2} - 10$

(%i2) `solve(%=10000,n);`

(%o2) $\left[n = -\frac{\sqrt{80081}+1}{2}, n = \frac{\sqrt{80081}-1}{2}\right]$

(%i3) `float(%[2]);`

(%o3) $n = 140.9929326856999$

(%i4) `sum(k,k,5,140);`

(%o4) 9860

(%i5) `sum(k,k,5,141);`

(%o5) 10001

Evidentemente la soluzione cercata è $n = 140$.

12.6 Confronto fra grafici

Questionario n. 4. Sessione ordinaria e supplettiva dell'Esame di Stato 2003-2004.

Risolvere la seguente disequazione in x: $(\ln x)^2 \geqslant \ln x^2$.

Facciamo un primo grafico (attenzione la scelta degli estremi non è casuale, ho fatto qualche prova) in modo da determinare gli intervalli solutivi e poi risolviamo l'equazione, è comunque ovvio che il dominio è $x > 0$.

(%i6) `plot2d([(log(x))^2,log(x^2)],[x,0,10],[y,-20,20],[gnuplot_preamble,"set zeroaxis;"])$`

(%i7)



Notiamo due intersezioni: α vicina a 1 e β compresa fra 7 e 8. Calcoliamole.

(%i7) `solve((log(x))^2=log(x^2),x);`

(%o1) $[x = 1, x = e^2]$

Dato che dobbiamo determinare gli intervalli per i quali $\log^2 x \geqslant \log x^2$ evidentemente le soluzioni sono: $]0,1]$ e $[e^2, +\infty[$.

12.7 Circuito in corrente continua

Determinare le correnti che fluiscono nel circuito rappresentato nella figura seguente.



Utilizzando le leggi di Kirchhoff si ottengono le seguenti equazioni:

$$\begin{cases} V_1 - I_1 R_1 - I_2 R_2 = 0 \\ V_2 - I_3 R_3 + I_2 R_2 = 0 \\ I_1 - I_2 - I_3 = 0 \end{cases}$$

Con Maxima utilizziamo il comando matrix e le operazioni fra matrici per risolvere il sistema lineare.

```
(%i1) M:matrix([R[1],R[2],0], [0,-R[2],R[3]], [-1,1,1]);
```

$$(\%o1) \quad \begin{pmatrix} R_1 & R_2 & 0 \\ 0 & -R_2 & R_3 \\ -1 & 1 & 1 \end{pmatrix}$$

```
(%i2) I:matrix([I[1]],[I[2]],[I[3]]);
```

$$(\%o2) \quad \begin{pmatrix} I_1 \\ I_2 \\ I_3 \end{pmatrix}$$

```
(%i3) E:matrix([E[1]],[E[2]],[0]);
```

$$(\%o3) \quad \begin{pmatrix} E_1 \\ E_2 \\ 0 \end{pmatrix}$$

```
(%i4) I=M^^(-1).E;
```

$$(\%o4) \quad \begin{pmatrix} I_1 \\ I_2 \\ I_3 \end{pmatrix} = \begin{pmatrix} \frac{E_1\,(R_3+R_2)}{(R_2+R_1)\,R_3+R_1\,R_2} + \frac{E_2\,R_2}{(R_2+R_1)\,R_3+R_1\,R_2} \\ \frac{E_1\,R_3}{(R_2+R_1)\,R_3+R_1\,R_2} - \frac{R_1\,E_2}{(R_2+R_1)\,R_3+R_1\,R_2} \\ \frac{E_2\,(R_2+R_1)}{(R_2+R_1)\,R_3+R_1\,R_2} + \frac{E_1\,R_2}{(R_2+R_1)\,R_3+R_1\,R_2} \end{pmatrix}$$

Si noti che se $E_1 = E_2$ la corrente $I_2 = 0$.

12.8 Scarica di un condensatore

La figura seguente mostra un circuito composto da una resistenza e un condensatore[8] t secondi dopo che è stato chiuso. Essendo stato carico il condensatore con una carica iniziale q_0, con la chiusura si genererà una corrente nel verso indicato che sarà presente finché non si giungerà ad uno stato di equilibrio.



L'equazione che descrive l'andamento della carica si deriva dalle leggi di Kirchhoff:

$$\frac{q(t)}{C} + \mathrm{RI} = 0$$

Ricordando che $I = \frac{\mathrm{dq}(t)}{\mathrm{dt}}$ l'equazione diventa:

$$\frac{q(t)}{C} + R\frac{\mathrm{dq}(t)}{\mathrm{dt}} = 0$$

Questa è una equazione differenziale ordinaria lineare del primo ordine che si risolverebbe facilmente con il metodo della separazione delle variabili. ma noi utilizzeremo l'istruzione ode2(eq,y,x) ove eq[9] è l'equazione da risolvere, y rappresenta la variabile dipendente e x quella indipendente.

```
(%i1) eq:q/C+R*'diff(q,t)=0;
```

(%o1) $\frac{d}{d\,t}\,q\,R + \frac{q}{C} = 0$

```
(%i2) ode2(eq,q,t);
```

(%o2) $q = \%\mathrm{c}\,\mathrm{e}^{-\frac{t}{C\,R}}$

```
(%i3) subst(0,t,%);
```

(%o3) $q = \%\mathrm{c}$

```
(%i4) subst(q[0],%c,%o2);
```

(%o4) $q = q_0\,\mathrm{e}^{-\frac{t}{C\,R}}$

Risolta l'equazione abbiamo posto le condizioni iniziali per determinare la costante $\%c$ che corrisponde alla carica iniziale q_0. Adesso che è nota la legge della carica possiamo determinare anche la legge della corrente ricordando che è la derivata della carica rispetto al tempo.

```
(%i5) esp:rhs(%o4);
```

(%o6) $q_0\,\mathrm{e}^{-\frac{t}{C\,R}}$

```
(%i7) I=diff(esp,t);
```

(%o7) $I = -\frac{q_0\,\mathrm{e}^{-\frac{t}{C\,R}}}{C\,R}$

8. Questo tipo di circuiti vengono chiamati RC, ove R sta per resistenza e C per condensatore.

9. Si faccia molta attenzione a come viene scritta l'equazione differenziale e all'uso dell'apice, per impedire a Maxima di derivare e avere una scrittura formale corretta.

Per avere una idea dell'andamento della carica e della corrente calcoliamo i limiti per $t \to \infty$.

(%i8) `limit(%o6,t,+inf);`

Is $C\,R$ positive or negative?`positive`

Is q_0 positive, negative, or zero?`positive`

(%o8) 0

(%i9) `limit(rhs(%o7),t,+inf);`

Is $C\,R$ positive or negative?`positive`

Is q_0 positive, negative, or zero?`positive`

(%o9) 0

Questi risultati confermano che la carica e la corrente fluiscono velocemente e, di fatto, dopo poco tempo si annullano. Il numero $\tau = \mathrm{RC}$ è chiamato tempo di dimezzamento e rappresenta il tempo dopo il quale un condensatore perde circa il 62% della carica[10].

Segue il grafico delle funzioni[11].



Esercizio 16. Utilizza gli stessi argomenti per risolvere il circuito RC con l'aggiunta di un generatore di tensione V_0. L'equazione del circuito diventerà $\frac{q(t)}{C} + \mathrm{RI} = V_0$, ecc. ecc.

13 Esercizi riassuntivi

13.1 Insiemi

Esercizio 17. Di quanti elementi è composto l'insieme $\mathcal{P}(\mathcal{P}(\mathcal{P}(\phi)))$?

10. Che è circa il 50%, appunto!
11. Abbiamo scelto $C = 5\mu F$, $R = 10\Omega$ e $q_0 = 12\mu C$.

13.2 Algebra

Esercizio 18. Valutare la seguente espressione per $x=1$ e $y=1$.

$$\left(y-\frac{x^2y}{y^2-x^2}\right)\frac{y-x}{y^2-2x^2}.$$

Esercizio 19. Riduci a forma nurmale il seguente monomio:

$$\left(-\frac{1}{3}\right)a^5b^2x(-5)a^{-2}\mathrm{bx}^3y.$$

Esercizio 20. Semplifica le seguenti espressioni algebriche:

a) $7x-5x+3x-2x-1$

b) $2a^2x+3a^2-7x+4a^2-a^2x-7a^2$

c) $6x+3y-9y+2+5x-2\mathrm{xy}$

d) $16\mathrm{ax}^3y^2:(-8\mathrm{xy}^2)$

e) $\mathrm{xy}^3:(-5\mathrm{ax}^3y^2)$

Esercizio 21. Calcolare il MCD e il mcm fra i seguenti monomi:
$2/3a^2b^2, a^3b^2x, 4/5a^2\mathrm{bx}^3y$

Esercizio 22. Calcola i seguenti prodotti notevoli:

a) $(2a-b^2/3)^2$

b) $(x+2y-1/2)^2$

c) $(x-1/2)^2$

d) $(a+3b-1)(a+3b+1)$

Esercizio 23. Calcola quoziente e resto delle seguente divisione:

$$x^4+5x^3+5x^2-5x-6:x+2$$

Esercizio 24. Fattorizza le seguenti espressioni:

a) $16a^4+b^4-8a^2b^2$

b) a^4-16b^4

c) $x^5+9\mathrm{xy}^2-6x^3y$

d) x^2-9y^2+4x+4

e) a^2+9a+8

f) x^3+3x^2-10x

Esercizio 25. Determina MCD e mcm fra i seguenti polinomi:
$4-4x^2, 1-x^3, 2x^2-4x+2$

Esercizio 26. Semplifica le seguenti frazioni algebriche:

a) $\left(\frac{a}{a-1}-\frac{2}{1-a^2}\frac{1}{a+1}\right):\frac{a^2-a-2}{a^2-2a+1}$

b) $\left(\frac{x+1}{2x}-\frac{x-1}{2x-1}\right)\frac{1-2x}{3x-1}-\left(1-\frac{1}{x}\right):\left(1+\frac{1}{x-2}\right)$

Esercizio 27. Risolvi le seguenti equazioni fratte e intere:

a) $\frac{2x}{4x^2-9}+\frac{3}{2x+3}=\frac{2}{x}$

b) $3(a-1)+2\mathrm{ax}=a^2-1-x(3-5a)$

Esercizio 28. In un rettangolo di perimetro 84cm la base è 3/4 dell'altezza. Calcola l'area del rettangolo.

Esercizio 29. Risolvi i seguenti sistemi lineari:

a) $\begin{cases} 5x/4+2y/3=7 \\ 3x-15+y=0 \end{cases}$

b) $\begin{cases} 2x-6y=9-6z \\ x+y-2z=0 \\ 4x-2z=5 \end{cases}$

13.3 Analisi

Esercizio 30. Calcolare i seguenti limiti

a) $\lim_{x\to 0}\frac{1-\cos x}{x}[=0]$

b) $\lim_{x\to 0} \frac{1-\cos x}{x^2}[=\frac{1}{2}]$

c) $\lim_{x\to 0} \frac{\tan x}{x}[=1]$

d) $\lim_{x\to 0} \frac{\arcsin x}{x}[=1]$

e) $\lim_{x\to +\infty} (\sqrt{x^2+1}-x)\sqrt{x^2+2}[=\frac{1}{2}]$

f) $\lim_{n\to +\infty} \left(1+\frac{1}{n^2}\right)^n[=+\infty]$

g) $\lim_{n\to +\infty} \left(1+\frac{1}{n}\right)^{n^2}[=1]$

Esercizio 31. Calcolare le derivate delle seguenti funzioni elementari (nella variabile x) semplificando fin dove possibile le espressioni ottenute.

a) $D_x\frac{1-x^2}{1+x^2}$

b) $D_x(1-2x)^3$

c) $D_x\mathrm{arcsin}(1-x^2)$

d) $D_x\sqrt{1+\sin^2 x}$

e) $D_x\mathrm{atan}\frac{x}{a}$

f) $D_x x^2 \ln x$

g) $D_x\ln(ax)$[12]

h) $D_x\mathrm{atan}\,\frac{2x}{1-x^2}$

i) $D_x\ln\left(x+\sqrt{1+x^2}\right)$

j) $D_x\sqrt{x}(1+\cos x)$

k) $D_x\ln\tan\frac{x}{2}$

Esercizio 32. Verificare che per $n=1,2,3,4$[13]

$$D_x^n\left(\frac{1}{a-x}\right)=\frac{n!}{(a-x)^{n+1}}$$

$$\frac{e^x}{n!}D_x^n(e^{-x}x^n)=\sum_{k=0}^{n}\frac{n!}{(n-k)!k!}(-1)^k\frac{x^k}{k!}=L_n(x)$$

I polinomi $L_n(x)$vengono detti polinomi di Laguerre.

Esercizio 33. Verificare mediante derivazione l'identità valida per $x>0$:

$$\mathrm{atan}\, x+\mathrm{atan}\frac{1}{x}=\frac{\pi}{2}$$

Esercizio 34. Calcolare i seguenti integrali indefiniti:

a) $\int x^2e^x\mathrm{dx}$

b) $\int \frac{\mathrm{dx}}{\sqrt{x}+1}$

c) $\int \mathrm{atan}\, x\mathrm{dx}$

d) $\int \tan^2 x\ \mathrm{dx}$

e) $\int e^{a\,x}\cos bx\mathrm{dx}$ (con a e b numeri reali)

f) $\int \frac{\mathrm{dx}}{x(1-x^3)}$

Esercizio 35. Verificare che se $|b|<a$ vale:

$$\int_o^\pi \frac{\mathrm{dx}}{a+b\cos x}=\frac{\pi}{\sqrt{a^2-b^2}}$$

Esercizio 36. Calcolare i seguenti integrali generalizzati:

a) $\int_0^1 \ln\frac{1}{x}\mathrm{dx}$

b) $\int_{-\infty}^{+\infty} \frac{\mathrm{dx}}{1+x^4}$

c) $\int_1^{+\infty} \frac{\mathrm{dx}}{x^2\ln x}$

12. Si noterà che la derivata non dipende dal parametro a. Come mai?

13. Ovviamente, visto che lavora Maxima, si può continuare.

Indice analitico

Bibliografia

[1] Enrico Centenaro. http://www.centenaro.net. sito dell'autore.

[2] D. E. Knuth. *The TEXbook*. Number ISBN 0-201-13447-0. Addison-Wesley, 1984.

[3] Leslie Lamport. *LATEX: A Document Preparation System*. Addison-Wesley, 1994.

[4] vari. http://maxima.sf.net. sito ufficiale.

[5] vari. http://www.texmacs.org. sito ufficiale.